

# Chapter 50

## Interactive Control Charts

### Chapter Table of Contents

---

<b>OVERVIEW</b> . . . . .	1791
<b>DETAILS</b> . . . . .	1792
Saving Graphics Coordinates in a Control Chart . . . . .	1792
Associating URLs with Subgroups in HTML . . . . .	1794
URLS and Tests for Special Causes . . . . .	1796



# Chapter 50

## Interactive Control Charts

---

### Overview

This chapter describes two approaches for creating an interactive control chart which allows an end user to “drill down” into subgroup data points and display information not contained in the chart itself. For example, the end user might want to be able to click on a subgroup to

- list the individual measurements in the subgroup
- diagnose an out-of-control point by viewing a Pareto chart of the most common problems affecting the process
- view a list of recommended corrective actions
- trace the raw materials used to manufacture a batch of product

The two approaches for creating interactive control charts are as follows:

- saving graphics coordinate data from control charts for use in creating SAS/AF applications
- associating Uniform Resource Locators (URLs) with subgroups to produce “clickable” control charts in HTML

The options described in this chapter can be specified in all the chart statements available in the SHEWHART procedure.

## Details

### Saving Graphics Coordinates in a Control Chart

You can specify an `WEBOUT=` data set in any chart statement to save graphics coordinate information for a control chart. The `WEBOUT=` data set is an extension of the `OUTTABLE=` data set, which contains the subgroup summary statistics, control limits and related information found in an `OUTTABLE=` data set, as well as coordinate data. The additional coordinate variables are listed in Table 50.1.

**Table 50.1.** `WEBOUT=` Data Set

Variable	Description
<code>_X1_</code>	x-coordinate of lower left corner of primary chart subgroup bounding box
<code>_Y1_</code>	y-coordinate of lower left corner of primary chart subgroup bounding box
<code>_X2_</code>	x-coordinate of upper right corner of primary chart subgroup bounding box
<code>_Y2_</code>	y-coordinate of upper right corner of primary chart subgroup bounding box
<code>_Xn_</code>	x-coordinate for point <i>n</i> of the subgroup shape
<code>_Yn_</code>	y-coordinate for point <i>n</i> of the subgroup shape
<code>_X1_2_</code>	x-coordinate of lower left corner of secondary chart subgroup bounding box
<code>_Y1_2_</code>	y-coordinate of lower left corner of secondary chart subgroup bounding box
<code>_X2_2_</code>	x-coordinate of upper right corner of secondary chart subgroup bounding box
<code>_Y2_2_</code>	y-coordinate of upper right corner of secondary chart subgroup bounding box
<code>_SHAPE_</code>	shape of primary chart subgroup bounding area
<code>_NXY_</code>	number of points defining primary chart subgroup bounding area
<code>_GRAPH_</code>	name of primary chart graphics entry
<code>_GRAPH2_</code>	name of secondary chart graphics entry
<code>_DXMIN_</code>	value of lowest major tick mark on horizontal axis
<code>_DXMAX_</code>	value of highest major tick mark on horizontal axis
<code>_XMIN_</code>	x-coordinate of lowest major tick mark on horizontal axis
<code>_XMAX_</code>	x-coordinate of highest major tick mark on horizontal axis
<code>_DYMIN_</code>	value of lowest major tick mark on vertical axis
<code>_DYMAX_</code>	value of highest major tick mark on vertical axis
<code>_YMIN_</code>	y-coordinate of lowest major tick mark on vertical axis
<code>_YMAX_</code>	y-coordinate of highest major tick mark on vertical axis
<code>_XMIN2_</code>	x-coordinate of lowest major tick mark on secondary chart horizontal axis
<code>_XMAX2_</code>	x-coordinate of highest major tick mark on secondary chart horizontal axis
<code>_DYMIN2_</code>	value of lowest major tick mark on secondary chart vertical axis
<code>_DYMAX2_</code>	value of highest major tick mark on secondary chart vertical axis
<code>_YMIN2_</code>	y-coordinate of lowest major tick mark on secondary chart vertical axis
<code>_YMAX2_</code>	y-coordinate of highest major tick mark on secondary chart vertical axis

You can use the coordinate data saved in the WEBOUT= data set to create a “clickable” control chart in a SAS/AF application. The variables `_X1_`, `_Y1_`, `_X2_` and `_Y2_` contain the coordinates of the lower left and upper right corners of a rectangular *bounding box* associated with each subgroup on the primary chart. This box defines the clickable area associated with the subgroup when the chart is incorporated into a SAS/AF application. It contains the symbol used to plot the subgroup data, or the junction of line segments representing the subgroup if no plotting symbol is used. The variables `_X1_2_`, `_Y1_2_`, `_X2_2_` and `_Y2_2_` contain coordinates of the corners of subgroup bounding boxes for a secondary chart.

If you use the BOXCHART statement, each subgroup is represented by a box-and-whisker plot rather than a single symbol. The subgroup’s bounding box is defined by the sides of the box-and-whisker plot and its lower and upper quartiles, regardless of the BOXSTYLE= value in effect.

If you specify the STARVERTICES= option, each subgroup is represented by a polygon or star with a vertex corresponding to each of the STARVERTICES= variables. The clickable area for a subgroup is the polygon with these vertices, regardless of the STARTYPE= value specified. In the WEBOUT= data set the value of the `_SHAPE_` variable is POLY and the `_NXY_` variable contains the number of vertices in the polygon. The variables `_Xn_` and `_Yn_`, where  $n = 1$  to the value of `_NXY_`, contain the coordinates of the vertices of a subgroup’s polygon. When the STARVERTICES= option is not used, the value of `_SHAPE_` is always RECT and the value of `_NXY_` is always 2.

When a control chart spans multiple panels (pages), the panels reside in separate SAS graphics entries. The `_GRAPH_` character variable records the name of the graphics entry containing the panel on which a given subgroup is plotted. This is the same name that appears in the PROC GREPLAY menu. When the SEPARATE option is used, primary and secondary charts are displayed on different graphics entries. The `_GRAPH2_` variable records the name of the graphics entry containing the secondary chart panel where a subgroup appears. When the SEPARATE option is not used, the values of `_GRAPH_` and `_GRAPH2_` will be the same for a given subgroup.

The variables `_DXMIN_`, `_DXMAX_`, `_XMIN_` and `_XMAX_` provide the data values and graphics coordinates associated with the lowest and highest major tick marks on the horizontal (subgroup) axis. The variables `_DYMIN_`, `_DYMAX_`, `_YMIN_` and `_YMAX_` provide the analogous values for the vertical axis. Through a simple linear transformation in your SAS/AF application you can use this information to convert from percent screen units to “data” units and vice versa.

The variables `_XMIN2_` and `_XMAX2_` contain the graphics coordinates associated with the lowest and highest major tick marks on the horizontal axis of a secondary chart. No variables for the corresponding data values are required, since they are always identical to those for the primary chart.

The variables `_DYMIN2_`, `_DYMAX2_`, `_YMIN2_` and `_YMAX2_` contain the data and coordinate values for the lowest and highest tick marks on the vertical axis of a secondary chart. A SAS/AF program receives the (x,y) coordinates for the location of the cursor when the user clicks on a subgroup data point. The application can

determine whether (x,y) lies within any of the boxes whose coordinates are saved in the WEBOUT= data set. If so, the program can determine which subgroup was selected on the primary or secondary chart and can check the \_TESTS\_ and \_TESTS2\_ variables included in the WEBOUT= data set to determine whether an out-of-control condition has been signaled.

**Notes:**

1. Graphics coordinates are scaled in percent screen units from 0 to 100, where (0,0) represents the lower-left corner of the screen and (100,100) represents the upper-right corner of the screen. Because SAS/AF applications define the origin of the vertical axis at the top of the screen, it will be necessary to subtract the y-coordinates from 100 in your SCL program.
2. The variables \_X1\_2\_, \_Y1\_2\_, \_X2\_2\_, \_Y2\_2\_, \_GRAPH2\_, \_XMIN2\_, \_XMAX2\_, \_YMIN2\_, \_YMAX2\_, \_DYMIN2\_ and \_DYMAX2\_ appear in the WEBOUT= data set only when a secondary chart is produced. A secondary chart is produced by the IRCHART, MRCHART, XRCHART and XSCHART statements and by the BOXCHART, MCHART and XCHART statements when the TRENDVAR= option is specified.
3. When the subgroup variable is a character variable, the value of \_DXMIN\_ is zero and the value of \_DXMAX\_ is the number of subgroups in the input data set minus one.
4. A bounding box circumscribes a point displayed on a chart and its dimensions depend on the size of the symbol marker used to display the point. If no symbol marker is specified, a small default size is used for the box. If a large number of subgroups are displayed on a panel, the subgroup symbols may overlap, so it is possible for a user to inadvertently select more than one point.

---

## Associating URLs with Subgroups in HTML

You can use the Output Delivery System (ODS) to produce an HTML file containing a control chart created by the SHEWHART procedure. The HTML= option provides a way to associate Uniform Resource Locators (URLs) with subgroups plotted on a control chart. It specifies a variable in the input data set whose values provide the URLs to be associated with different subgroups. The HTML= variable can be a character variable or a numeric variable with an associated character format.

The following statements generate an  $\bar{X}$  chart that is saved to a GIF file and included in an HTML file. The formatted values of the numeric HTML= variable WEB are URLs that link subgroups in the input data set to various web pages.

```

options target = gif;
ods html body = "example1.html";

proc format;
  value webfmt
    1='href="http://www.sas.com/'
    2='href="http://www.sas.com/service/techsup/faq/qc/shewproc.html"'
    3='href="http://www.sas.com/rnd/app/qc.html"'
    4='href="http://www.sas.com/rnd/app/qc/qcnew.html"'
    5='href="http://www.sas.com/software/web_enablement/'
  ;

data wafers;
  format web webfmt.;
  input batch web @;
  do i=1 to 5;
    input diamtr @;
    output;
  end;
  drop i;
datalines;
1 1 35.00 34.99 34.99 34.98 35.00
2 1 35.00 34.99 34.99 34.98 35.00
3 1 34.99 34.99 35.00 34.99 35.00
4 1 35.00 35.00 34.99 34.99 35.00
5 2 35.00 34.99 34.98 34.99 35.00
6 2 34.99 34.99 35.00 35.00 35.00
7 2 35.01 34.98 35.00 35.00 34.99
8 2 35.00 35.00 34.99 34.98 34.99
9 3 34.99 34.98 34.99 35.01 35.00
10 3 34.99 35.00 35.00 34.99 35.00
11 3 35.01 35.00 35.00 34.98 34.99
12 3 34.99 34.99 35.00 34.98 35.01
13 4 35.01 34.99 34.98 34.99 34.99
14 4 35.00 35.00 34.99 35.00 34.99
15 4 34.98 35.00 34.99 35.00 34.99
16 4 34.99 35.00 35.00 35.01 35.00
17 5 34.98 34.98 34.98 34.99 34.98
18 5 35.01 35.02 35.00 34.98 35.00
19 5 34.99 34.98 35.00 34.99 34.98
20 5 34.99 35.00 35.00 34.99 34.99
;

symbol1 v=square;
proc shewhart data=wafers;
  xchart diamtr*batch / html = ( web );
run;

ods html close;
run;

```

In this example five different URLs are each associated with a set of four subgroup values. When you view the ODS HTML output with a browser, you can click on

a subgroup data point and the browser will bring up the page specified by the subgroup's URL. These URLs happen to point to pages at SAS Institute's web site which may be of interest to SAS/QC users.

**Note:** The value of the HTML= variable must be the same for each observation belonging to a given subgroup.

---

## URLS and Tests for Special Causes

The TESTURLS= data set provides a way to associate a URL with each subgroup in a control chart for which a given test for special causes is positive:

**Table 50.2.** Variables Required in a TESTURLS= Data Set

Variable	Type	Description
_TEST_	character or numeric	test identifier
_CHART_	numeric	primary (1) or secondary (2) chart
_URL_	character	URL associated with subgroups with positive test

The variable \_TEST\_ identifies a test for special causes (see Chapter 48, "Tests for Special Causes" on page 1731). A standard test is identified by its number (1 to 8) and a nonstandard test is identified by the CODE= character in its pattern specification. The \_TEST\_ variable must be a character variable if nonstandard tests are included in the TESTURLS= data set. The value of \_CHART\_ is 1 or 2, specifying whether the test applies to the primary or secondary chart. The character variable \_URL\_ contains the URL link to be associated with subgroups for which the test is positive.

The following statements create a TESTURLS= data set and an  $\bar{X}$  chart using the same DATA= data set as the previous example:

```
ods html body = "example2.html";

data testlink;
    length _URL_ $ 75;
    input _TEST_ _CHART_ _URL_;
datalines;
1 1 href="http://www.sas.com/"
2 1 href="http://www.sas.com/service/techsup/faq/qc/shewproc.html"
3 1 href="http://www.sas.com/rnd/app/qc.html"
4 1 href="http://www.sas.com/rnd/app/qc/qcnew.html"
5 1 href="http://www.sas.com/software/web_enablement/"
6 1 href="http://www.sas.com/rnd/app/qc/qcspc.html"
7 1 href="http://www.sas.com/software/components/qc.html"
8 1 href="http://www.sas.com/rnd/app/qc/qcover.html"
;
```



```
symbol1 v=dot;  
proc shewhart data=wafers testurls=testlink;  
  xchart diamtr*batch / tests = 1 to 8;  
run;  
  
ods html close;  
run;
```

In this example only subgroups triggering tests for special causes have URLs associated with them.

**Note:** If a TESTURLS= data set and an HTML= variable are both specified, the URL from the TESTURLS= data set is associated with any subgroup for which the test is positive.

The correct bibliographic citation for this manual is as follows: SAS Institute Inc., *SAS/QC<sup>®</sup> User's Guide, Version 8*, Cary, NC: SAS Institute Inc., 1999. 1994 pp.

**SAS/QC<sup>®</sup> User's Guide, Version 8**

Copyright © 1999 SAS Institute Inc., Cary, NC, USA.

ISBN 1-58025-493-4

All rights reserved. Printed in the United States of America. No part of this publication may be reproduced, stored in a retrieval system, or transmitted, by any form or by any means, electronic, mechanical, photocopying, or otherwise, without the prior written permission of the publisher, SAS Institute Inc.

**U.S. Government Restricted Rights Notice.** Use, duplication, or disclosure of the software by the government is subject to restrictions as set forth in FAR 52.227-19 Commercial Computer Software-Restricted Rights (June 1987).

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

1st printing, October 1999

SAS<sup>®</sup> and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute in the USA and other countries. <sup>®</sup> indicates USA registration.

IBM<sup>®</sup>, ACF/VTAM<sup>®</sup>, AIX<sup>®</sup>, APPN<sup>®</sup>, MVS/ESA<sup>®</sup>, OS/2<sup>®</sup>, OS/390<sup>®</sup>, VM/ESA<sup>®</sup>, and VTAM<sup>®</sup> are registered trademarks or trademarks of International Business Machines Corporation. <sup>®</sup> indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.

The Institute is a private company devoted to the support and further development of its software and related services.